Коллеги, помогите понять, где проблема
Есть запрос:

ВЫБРАТЬ
 ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки,
 ХарактеристикаСырьяМесторожденияХарактеристики.Харка,
 ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin,
 ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMax,
 ХарактеристикаСырьяМесторожденияХарактеристики.Среднее,
 ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка,
 ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка.ПринадлежностьМесторождению
ИЗ
 Документ.ХарактеристикаСырьяМесторождения.Характеристики КАК ХарактеристикаСырьяМесторожденияХарактеристики
ГДЕ
 ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка.Сырье = &Сырье
 И (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки1
 И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка1
 И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin > &ЗначениеMin1)
 И (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки2
 И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка2
 И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin < &ЗначениеMin2)

Делаю выборку из документов, где в табличной части есть совпадающие условия
Если убрать второе "И"

( И (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки2
 И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка2
 И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin < &ЗначениеMin2))

Запрос корректно работает, а если не убирать - пишет, что ничего не найдено.
Все значения проверил, такой документ существует, но запрос не отрабатывает
---
psa247,
даже наверное точнее вот как то так, на примере другого типа документа, свои поля просто подставишь

```1c
    ВЫБРАТЬ
    Вложение.Ссылка
    ИЗ
    (ВЫБРАТЬ
    ПоступлениеТоваровУслугТовары.Ссылка КАК Ссылка,
    1 КАК Поле1,
    0 КАК Поле2
    ИЗ
    Документ.ПоступлениеТоваровУслуг.Товары КАК ПоступлениеТоваровУслугТовары
    ГДЕ
    ПоступлениеТоваровУслугТовары.Ссылка.Склад = &Склад
    И ПоступлениеТоваровУслугТовары.Номенклатура = &Номенклатура1
    И ПоступлениеТоваровУслугТовары.Характеристика = &Характеристика1
    И ПоступлениеТоваровУслугТовары.Цена > &Цена1

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    ПоступлениеТоваровУслугТовары.Ссылка,
    0,
    1
    ИЗ
    Документ.ПоступлениеТоваровУслуг.Товары КАК ПоступлениеТоваровУслугТовары
    ГДЕ
    ПоступлениеТоваровУслугТовары.Ссылка.Склад = &Склад
    И ПоступлениеТоваровУслугТовары.Номенклатура = &Номенклатура2
    И ПоступлениеТоваровУслугТовары.Характеристика = &Характеристика2
    И ПоступлениеТоваровУслугТовары.Цена < &Цена2) КАК Вложение

    СГРУППИРОВАТЬ ПО
    Вложение.Ссылка

    ИМЕЮЩИЕ
    СУММА(Вложение.Поле1) > 0 И
    СУММА(Вложение.Поле2) > 0
```
---
psa247,
а что такое Харка?
---
AlexO, Составной тип из справочников
---
psa247, оставной тип из справочников
новый тип в 1С? :)
---
где я туплю ? )))
---
psa247, одна Харка не может одновременно и =&Харка1 и =&Харка2, ты уж определись: чему она должна равняться, с ТипХарка кстати тоже
---
Скорее всего, Вы правы. Но как выбрать документы, у которых в табличной части есть Харка1 и ее значение Мин1
и Харка2, где ее значение, к примеру, больше 10 ??
---
Соединением ?
---
[CODE
ГДЕ
ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка.Сырье = &Сырье
И
       (
         (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки1
          И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка1
          И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin > &ЗначениеMin1)
       ИЛИ
         (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки2
          И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка2
          И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin < &ЗначениеMin2)
       )
[/1C-CODE]
---
Так пойдет только в одном случае, если клиента интересует ИЛИ это ИЛИ то.
Попробую вложенных запросов понаделать и объединить )))
---
psa247, в одной строчке ТЧ документа одно поле Харка?
---
m-serg74, Да, одно
---
psa247, тогда я так понимаю надо выбрать документы где есть одновременно две строки одна из которых подходит под условие:

```1c
    (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки1
    И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка1
    И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin > &ЗначениеMin1)
```

и вторая под условие:

```1c
    (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки2
    И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка2
    И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin < &ЗначениеMin2)
    )
```

но тогда не получится:

```1c
    ВЫБРАТЬ
    ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки,
    ХарактеристикаСырьяМесторожденияХарактеристики.Харка,
    ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin,
    ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMax,
```

а лишь только получить ссылку на сам документ...
---
Да, достаточно ссылку на документ получить
Убрал лишнее, запрос получается такой:

ВЫБРАТЬ
 ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка
ИЗ
 Документ.ХарактеристикаСырьяМесторождения.Характеристики КАК ХарактеристикаСырьяМесторожденияХарактеристики
ГДЕ
 ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка.Сырье = &Сырье
 И (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки1
 И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка1
 И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin > &ЗначениеMin1)
 И (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки2
 И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка2
 И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin < &ЗначениеMin2)

Но все равно пустой результат
---
psa247,

```1c
    ВЫБРАТЬ
    Вложение.Ссылка
    ИЗ
    (ВЫБРАТЬ
    ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка КАК Ссылка,
    СУММА(1) КАК Признак
    ИЗ
    Документ.ХарактеристикаСырьяМесторождения.Характеристики КАК ХарактеристикаСырьяМесторожденияХарактеристики
    ГДЕ
    ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка.Сырье = &Сырье
    И
    (
    (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки1
    И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка1
    И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin > &ЗначениеMin1)
    ИЛИ
    (ХарактеристикаСырьяМесторожденияХарактеристики.ТипХарки = &ТипХарки2
    И ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка2
    И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin < &ЗначениеMin2)
    )
    СГРУППИРОВАТЬ ПО
    ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка) КАК Вложение
    ГДЕ
    Вложение.Признак >= 2
```
---
psa247,
даже наверное точнее вот как то так, на примере другого типа документа, свои поля просто подставишь

```1c
    ВЫБРАТЬ
    Вложение.Ссылка
    ИЗ
    (ВЫБРАТЬ
    ПоступлениеТоваровУслугТовары.Ссылка КАК Ссылка,
    1 КАК Поле1,
    0 КАК Поле2
    ИЗ
    Документ.ПоступлениеТоваровУслуг.Товары КАК ПоступлениеТоваровУслугТовары
    ГДЕ
    ПоступлениеТоваровУслугТовары.Ссылка.Склад = &Склад
    И ПоступлениеТоваровУслугТовары.Номенклатура = &Номенклатура1
    И ПоступлениеТоваровУслугТовары.Характеристика = &Характеристика1
    И ПоступлениеТоваровУслугТовары.Цена > &Цена1

    ОБЪЕДИНИТЬ ВСЕ

    ВЫБРАТЬ
    ПоступлениеТоваровУслугТовары.Ссылка,
    0,
    1
    ИЗ
    Документ.ПоступлениеТоваровУслуг.Товары КАК ПоступлениеТоваровУслугТовары
    ГДЕ
    ПоступлениеТоваровУслугТовары.Ссылка.Склад = &Склад
    И ПоступлениеТоваровУслугТовары.Номенклатура = &Номенклатура2
    И ПоступлениеТоваровУслугТовары.Характеристика = &Характеристика2
    И ПоступлениеТоваровУслугТовары.Цена < &Цена2) КАК Вложение

    СГРУППИРОВАТЬ ПО
    Вложение.Ссылка

    ИМЕЮЩИЕ
    СУММА(Вложение.Поле1) > 0 И
    СУММА(Вложение.Поле2) > 0
```
---
m-serg74, Спасибо. Ответ на мысль навел - сделаю Условия в Группе
---
psa247, не совсем понял... про "Условия в Группе", можешь показать?
---
Хорош прикалываться
---
ВЫБРАТЬ
 ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка,
 ХарактеристикаСырьяМесторожденияХарактеристики.Харка,
 ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin
ИЗ
 Документ.ХарактеристикаСырьяМесторождения.Характеристики КАК ХарактеристикаСырьяМесторожденияХарактеристики
ГДЕ
 ХарактеристикаСырьяМесторожденияХарактеристики.Ссылка.Сырье = &Сырье
 И ВЫБОР
   КОГДА ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка1
    ТОГДА ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin > &ЗначениеMin1
      И ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin <> 0
   КОГДА ХарактеристикаСырьяМесторожденияХарактеристики.Харка = &Харка2
    ТОГДА ХарактеристикаСырьяМесторожденияХарактеристики.ЗначениеMin < &ЗначениеMin2
   ИНАЧЕ ЛОЖЬ
  КОНЕЦ
---
Ну и, где надо - вставляю ИЛИ с таким же набором условий
---
psa247, не факт что получишь документы в которых обязательно присутствуют хотя бы две строки попадающие под два нужных условия, все равно как в (13) придется группировать с контролем двух видов строк, так сказать, я вчера сначала сделал (12), а потом понял что, если попадутся две строки по Условию1, то такой документ все равно выведется, а тебе он не нужен, поэтому и переделал на (13)
---
m-serg74, спасибо. проверю
---
psa247, удачи!
---
m-serg74, точно, вернусь к варианту ОБЪЕДИНИТЬ ВСЕ, еще раз спасибо ))